TOC Assignment 2

Akshat Oke 2020A7PS0284H
Aaditya Rathi 2020A7PS2191H
Sankalp Kulkarni 2020A7PS1097H
Shreyas Dixit 2020A7PS2079H
Chirag Gadia 2020A7PS1721H

December 2022

Contents

1 Grammar for Simple C

1.1 Grammar syntax

The grammar is mostly in BNF, with the following qualifiers that may follow a grouping or a terminal/non-terminal.

+	One or more of target
*	Zero or more of target
?	Zero or one of target

A grouping of symbols is enclosed by brackets '[' and ']' or '(' and ')'. Additionally, '-' represents negation, which shall *not* match the symbol following it.

[0..9] represents the range of characters 0,1,2,3,4,5,6,7,8,9, similarly for [A..Z] and [a..z].

1.2 The Grammar

Start symbol is ‘MainProgram‘

⟨*MainProgram*⟩ ::= ⟨*Declarations*⟩
| ⟨*Program*⟩
| ⟨*Declarations*⟩ ⟨*Program*⟩

⟨*Declarations*⟩ ::= ‘`int`’ ⟨*identifier*⟩ [‘`,`’ ⟨*identifier*⟩]* ‘`;`’

⟨*Program*⟩ ::= ⟨*statement*⟩+

⟨*statement*⟩ ::= (⟨*assignment*⟩
| ⟨*read-statement*⟩
| ⟨*write-statement*⟩
| ⟨*for-statement*⟩
) ‘`;`’

⟨*assignment*⟩ ::= ⟨*identifier*⟩ ‘`=`’ ⟨*expression*⟩

⟨*expr-or-assign*⟩ ::= ⟨*expression*⟩ [‘`=`’ ⟨*expression*⟩]?

⟨*expression*⟩ ::= ⟨*equality*⟩

⟨*equality*⟩ ::= ⟨*comparison*⟩ [‘`==`’ ⟨*comparison*⟩]?

⟨*comparison*⟩ ::= ⟨*term*⟩ [‘`>`’ ⟨*term*⟩]?

⟨*term*⟩ ::= ⟨*factor*⟩ [(‘`+`’ | ‘`-`’) ⟨*factor*⟩]*

⟨*factor*⟩ ::= ⟨*unary*⟩ [(‘`*`’ | ‘`/`’) ⟨*unary*⟩]*

⟨*unary*⟩ ::= ⟨*primary*⟩

⟨*primary*⟩ ::= ⟨*integer*⟩ | ⟨*identifer*⟩ | ‘`(`’ ⟨*expression*⟩ ‘`)`’

⟨*read-statement*⟩ ::= ‘`read`’ ⟨*identifier*⟩

⟨*write-statement*⟩ ::= ‘`write`’ (⟨*integer*⟩ | ⟨*identifer*⟩)

⟨*for-statement*⟩ ::= '`for`' '`(`' ⟨*expr-or-assign*⟩? '`;`' ⟨*expression*⟩? '`;`' ⟨*expr-or-assign*⟩? '`)`' ⟨*block*⟩

⟨*block*⟩ ::= '`{`' ⟨*statement*⟩* '`}`'

⟨*identifier*⟩ ::= - ⟨*keyword*⟩ ⟨*alpha*⟩+

⟨*integer*⟩ ::= ⟨*digit*⟩+

⟨*digit*⟩ ::= [0..9]

⟨*alpha*⟩ ::= [A..Z] | [a..z]

1.3 Comments on the grammar

It is possible to skip the <equality> rule as it only forwards <expression>, but makes the meaning clearer.

The grammar is not left-associative as required, for it would require it to be left-recursive as well. Since the parsing algorithm used is recursive descent, the additive and multiplicative productions are flattened out and interpreted left-recursively by the interpreter.

Just replacing the Kleene star by '+' in <comparison> and <equality> would have resulted in Python-style expressions such as `a>b>c`!

2 About the implementation

2.1 Types

Since the document mentions nothing about types, this implementation assumes 'Simple C' to be strongly typed, with two types.

There are two types of expressions: number and boolean. Only a boolean expression can be used in the condition clause of the for statement.
Printing a boolean variable will print 'true' or 'false' correspondingly. Booleans cannot be operated upon by arithmetic operations.

2.2 Uninitialized variables

The <Declarations> rule declares variables that can be used in the program. Variables have been implemented with a hashmap. Referencing a variable before assigning a value to it causes an 'Uninitialized variable' runtime error.

It was possible to implicity give variables an initial value of 0, but this may lead to unclear code.

2.3 Ghost nodes

The for statement grammar may not include all three clauses, for example `for(;a<4;)`. In such cases the parser inserts a 'ghost node' which evaluates to 'true' but does not get printed by the AST printer code.

This ghost node is also used in <expr-or-assign> in order to provide easy access to the lvalue identifier lexeme.

3 Remarks

Before the correction in the for statement syntax, all three clauses were expressions and had no side effects. This means they could be executed any number of times and in any order during the loop execution. Although it seemed that I could get away easily by not using the AST for interpreting, the additional "rewind" functionality required in the scanner was better left alone.